

STICHTING
MATHEMATISCH CENTRUM
2e BOERHAAVESTRAAT 49
AMSTERDAM

AFDELING TOEGEPASTE WISKUNDE


TN 45

Algebraic operations in ALGOL 60

(The saddlepoint method)

by

S.J. Bijlsma



May 1966

The Mathematical Centre at Amsterdam, founded the 11th of February, 1946, is a non-profit institution aiming at the promotion of pure mathematics and its applications, and is sponsored by the Netherlands Government through the Netherlands Organization for the Advancement of Pure Research (Z.W.O.) and the Central Organization for Applied Scientific Research in the Netherlands (T.N.O.), by the Municipality of Amsterdam and by several industries.

1. Saddle point method


In the following we want to study the asymptotic behaviour of $\phi(s)$ defined by:

$$(1) \quad \phi(s) = \int_c e^{-sf(z)} g(z) dz$$

for large positive values of s , assuming that $f(z)$ and $g(z)$ ($f(z) = R(x,y) + iI(x,y)$) are analytic functions of a complex variable z and $R(x,y) \rightarrow +\infty$ at both ends of the contour c . We consider a complex transformation from the z - to the w - ($= u + iv$) plane:

$$\begin{aligned} w &= f(z) \quad \text{or:} \\ u + iv &= R(x,y) + iI(x,y) \end{aligned}$$

This transformation is singular at a point z_0 , where $f'(z_0) = 0$ (saddle point) and gives rise to the branchpoint w_0 in the w -plane. Let c' be the transformed contour c as is illustrated in the figure:



then we have:

$$(2) \quad \phi(s) = \int_{c'} e^{-sw} g(z(w)) \frac{dz}{dw} \cdot dw.$$

The idea of the saddlepoint method is to reduce c' into such a path that $|e^{-sw}| = e^{-su}$ becomes as small as possible on this path, i.e., c' is reduced into c'' .

In the following we assume that z_0 is a first order saddlepoint ($f''(z_0) \neq 0$). By the local substitution $(w - w_0)^{\frac{1}{2}} = p$, (2) changes into:

$$(3) \quad e^{-sf(z_0)} \int_{-\infty}^{\infty} e^{-sp^2} g(z(p)) \left(\frac{dz}{dw} \circ \frac{dw}{dp} \right) dp.$$

2. Short description of the ALGOL-programs

In order to calculate the asymptotic series, we first determine by means of program SJB 040166/1 the zero of the derivative of the complex function $f(z)$. Then program SJB 040166/2 is used to invert the series $f(z) - f(z_0) = p^2$, to substitute the found series in $g(z(p)) \frac{dz}{dp}$ and to calculate the series in p , after which one can integrate formula (3) in a elementary way.

To be more specific, let:

$$\frac{f''(z_0)}{2!} (z - z_0)^2 + \dots + \frac{f^n(z_0)}{n!} (z - z_0)^n + \dots = p^2.$$

Suppose we put:

$$z - z_0 = \sum_{i=0}^{\infty} a_i p^{i+1} \quad \text{and} \quad \frac{f^k(z_0)}{k!} = b_k$$

then we can find the coefficients a_i by comparison of coefficients.

This gives:

$$b_2 \left(\sum_{i=0}^{\infty} a_i p^i \right)^2 + \dots + b_n \left(\sum_{i=0}^{\infty} a_i p^i \right)^n p^{n-2} + \dots = 1.$$

As a result we find:

$$a_0 = \sqrt{\frac{1}{b_2}};$$

let the coefficient of p^j in $\left(\sum_{i=0}^{\infty} a_i p^i \right)^l$ be $som(l, j)$ then we have

moreover

$$a_n = \frac{-\left\{ \sum_{k=3}^{n+2} b_k \cdot \text{som}(k, n-k+2) + \sum_{i=1}^{n-1} a_i \cdot a_{n-i} \right\}}{2a_0}$$

where if $l = 1$: $\text{som}(l, j) = a_j$, else if $l > 1$ then:

$$\text{som}(l, j) = \sum_{i=0}^j a_i \cdot \text{som}(l-1, j-i).$$

If: $g(z) = d_0 + d_1(z - z_0) + \dots + d_n(z - z_0)^n + \dots$

then the coefficient e_k in:

$$g(z(p)) = e_0 + e_1 p + \dots + e_n (z - z_0)^n + \dots$$

is given by

$$e_k = \sum_{i=1}^k d_i \cdot \text{som}(i, k-i) \quad \text{and}$$

$$e_0 = d_0.$$

Assuming that:

$$g(z(p)) \frac{dz}{dp} = g_0 + g_1 p + \dots + g_n \cdot p^n + \dots$$

we have:

$$g_k = \sum_{i=0}^k e_{k-i} \cdot (i+1) a_i$$

so that formula (3) changes into:

$$e^{-sf(z_0)} \int_{-\infty}^{\infty} e^{-sp^2} (g_0 + \dots + g_n p^n + \dots) dp$$

and as a result we have:

$$e^{-sf(z_0)} \sqrt{\frac{\pi}{s}} (g_0 + \dots + \frac{g_{2n}(3 \cdot 5 \cdot \dots \cdot (2n-1))}{s^n \cdot 2^n} + \dots)$$

The coefficients g_k are given by program SJB 040166/2.

3. Input

The formulas $f(z)$ and $g(z)$, written in Polish notation are punched in the input paper tape. The operation symbols and function symbols are punched as integers according to the following table

*	+	/	-	sin	cos	exp	ln	sqrt	arctg
1	2	3	4	5	6	7	8	9	10

The variable z is punched as the integer 11. A complex number c is punched as the integer 12, followed by two real numbers respectively equal to the real and imaginary part of c . Example:

The formula $\frac{z \times z}{2} + \ln z$, in Polish notation $+ / * z z 2 \ln z$ is represented on the tape by the following sequence of numbers:

2 3 1 11 11 12 2 0 8 11

Input of SJB 040166/1 consists of 5 real numbers $\bar{x}_0, \bar{y}_0, \epsilon_1, \epsilon_2, \epsilon_3$ and the formula $f(z)$, where $\bar{x}_0 + i\bar{y}_0$ is a first approximation of the zero z_0 and $\epsilon_1, \epsilon_2, \epsilon_3$ are desired accuracies (see procedure COMPLEX ZERO). Output of this program consists of 2 real numbers x_0 and y_0 , where $x_0 + iy_0$ is the calculated value of the zero z_0 , with the above mentioned accuracy.

Input of SJB 040166/2 consists of the integer m_0 , the real numbers x_0 and y_0 (output SJB 040166/1), the formulas $f(z)$ and $g(z)$, where $(m_0 - 2)$ is the index of the last of the coefficients g_k . Output of this program are the calculated coefficients g_k represented as a pair of real numbers respectively equal to the real and imaginary part of g_k .

The following M.C. Standardprocedures are used:

COMPLEX ZERO (AP 217): A procedure giving the zero of a function of a complex variable.

XEEN(n) : An integer procedure assigning to its identifier a number which can be brought into the machine by the console and which is used here to determine the array-lengths H and HC , in which formulas respectively complex numbers are stored.

Acknowledgement

The idea of manipulating formulas, as used in this note, is due to R.P. v.d. Riet (ref. 2 and 3.).

```
begin comment Opdrachtnr R1261 / TW160. Codenr SJB 040166 / 1.
      Bepaling reele en imaginaire delen;
integer kmax, kcmx; kmax:=XEEN(1023 ); kcmx:=XEEN(1023×1024): 1024 ;
begin integer i,k,kc,x,y,z,R,I; real x0,y0,X0,Y0;
      integer array H[0:kmax,1:3]; real array HC[0:kcmx,1:2], e[1:3];
integer procedure STORE (i,l,j); value i,j; integer i,l,j;
begin STORE:=k:=k+1; if k>kmax then begin PUTEXT1(⌊ k too large⌋);stop end;
      H[k,1 ]:=i;H[k,2 ]:=1;H[k,3 ]:=j
end;
integer procedure S(i,j); value i,j; integer i,j; S:=
if i=0 then j else if j=0 then i else STORE(i,1,j);
integer procedure D(i,j); integer i,j; D:=S(i,P(NUMBER(STORECN(-1,0)),j));
integer procedure P(i,j); value i,j; integer i,j; P:=
if i=0∨ j=0 then 0 else if i=1 then j else if j=1 then i else STORE (i,2,j);
integer procedure Q(i,j); integer i,j; Q:= STORE (i,3,j);
integer procedure SIN(i); integer i; SIN:=STORE(1,0,i);
integer procedure COS(i); integer i; COS:=STORE(2,0,i);
integer procedure EXP (i); integer i; EXP:=STORE (3,0,i);
integer procedure LN(i); integer i; LN:= STORE (4,0,i);
integer procedure SQRT(i); integer i; SQRT:=STORE (5,0,i);
integer procedure ARCTAN(i); integer i; ARCTAN:=STORE (6,0,i);
integer procedure NUMBER(i); integer i; NUMBER:=STORE(-1,0,i);
integer procedure STORECN(a1,a2);real a1,a2;
begin STORECN:=kc:=kc+1;
      if kc>kcmx then begin PUTEXT1(⌊kc too large⌋);stop end;
      HC[kc,1 ]:=a1; HC[kc,2 ]:=a2
end;
procedure RI(i,R,I);value i; integer i,R,I;
begin integer r1,r2,i1,i2;
if i=z then begin R:=x;I:=y end else
if H[i,1 ]=-1 then begin R:=NUMBER(STORECN(HC[H[i,3 ],1 ],0));
      I:= NUMBER(STORECN(0,HC[H[i,3 ],2 ]))
      end
      else
if H[i,2 ]=1 then begin RI(H[i,1 ],r1,i1);RI(H[i,3 ],r2,i2); R:=S(r1,r2);
      I:= S(i1,i2)
```

```

                                end
                                else
if H[i,2]=2 then begin RI(H[i,1],r1,i1);RI(H[i,3],r2,i2);
                                R:=D(P(r1,r2),P(i1,i2)); I:=S(P(r1,i2),P(r2,i1))
                                end
                                else
if H[i,2]=3 then begin RI(H[i,1],r1,i1);RI(H[i,3],r2,i2);
                                R:=Q(S(P(r1,r2),P(i1,i2)),S(P(r2,r2),P(i2,i2)));
                                I:=Q(D(P(i1,r2),P(r1,i2)),S(P(r2,r2),P(i2,i2)))
                                end
                                else
if H[i,1]=3  $\wedge$  H[i,2]=0 then begin RI(H[i,3],r1,i1);
                                R:=P(EXP(r1),COS(i1));
                                I:=P(EXP(r1),SIN(i1))
                                end
                                else
if H[i,1]=2  $\wedge$  H[i,2]=0 then begin RI(H[i,3],r1,i1);
                                R:=P(P(NUMBER(STORECN(.5,0)),COS(r1)),
                                S(EXP(i1),Q(1,EXP(i1))));
                                I:=P(P(NUMBER(STORECN(.5,0)),SIN(r1)),
                                D(Q(1,EXP(i1)),EXP(i1)))
                                end
                                else
if H[i,1]=1  $\wedge$  H[i,2]=0 then begin RI(H[i,3],r1,i1);
                                R:=P(P(NUMBER(STORECN(.5,0)),SIN(r1)),
                                S(EXP(i1),Q(1,EXP(i1))));
                                I:=P(P(NUMBER(STORECN(-.5,0)),COS(r1)),
                                D(Q(1,EXP(i1)),EXP(i1)))
                                end
                                else
if H[i,1]=4  $\wedge$  H[i,2]=0 then begin RI(H[i,3],r1,i1);
                                R:=LN(SQRT(S(P(r1,r1),P(i1,i1))));
                                I:=ARCTAN(Q(i1,r1))
                                end
                                else
```

```

if H[i,1] = 5 ∧ H[i,2] = 0 then begin RI(H[i,3],r1,i1);
    R:=P(SQRT(S(P(r1,r1),P(i1,i1))),
    COS(P(ARCTAN(Q(i1,r1)),
    NUMBER(STORECN(.5,0))));
    I:=P(SQRT(S(P(r1,r1),P(i1,i1))),
    SIN(P(ARCTAN(Q(i1,r1)),
    NUMBER(STORECN(.5,0)))))
    end
else
if H[i,1] = 6 ∧ H[i,2] = 0 then begin RI(H[i,3],r1,i1);
    R:=P(NUMBER(STORECN(.5,0)),
    D(ARCTAN(Q(P(NUMBER(STORECN(-1,0)),r1),
    S(i1,NUMBER(STORECN(-1,0))))) ,
    ARCTAN(Q(P(NUMBER(STORECN(-1,0)),r1),
    S(i1,1))))) );
    I:=P(NUMBER(STORECN(-.5,0)),
    LN(SQRT(Q(S(P(r1,r1),P(D(1,i1),D(1,i1))),
    S(P(r1,r1),P(S(1,i1),S(1,i1))))) )
    end
end;
real procedure VALUECN(i); value i; integer i;
VALUECN:=if abs(HC[H[i,3],1]) < 10 then HC[H[i,3],2] else HC[H[i,3],1];
integer procedure DI(i); value i; integer i; DI:=
if H[i,1] = -1 then 0 else if i = z then 1 else
if H[i,2] = 1 then S(DI(H[i,1]),DI(H[i,3])) else
if H[i,2] = 2 then S(P(H[i,1],DI(H[i,3])),P(H[i,3],DI(H[i,1]))) else
if H[i,2] = 3 then Q(D(P(H[i,3],DI(H[i,1])),P(H[i,1],DI(H[i,3]))),
    P(H[i,3],H[i,3]))
    else
if H[i,1] = 1 ∧ H[i,2] = 0 then P(COS(H[i,3]),DI(H[i,3])) else
if H[i,1] = 2 ∧ H[i,2] = 0 then P(NUMBER(STORECN(-1,0)),
    P(SIN(H[i,3]),DI(H[i,3])))
    else
if H[i,1] = 3 ∧ H[i,2] = 0 then P(EXP(H[i,3]),DI(H[i,3])) else
if H[i,1] = 4 ∧ H[i,2] = 0 then P(Q(1,H[i,3]),DI(H[i,3])) else
if H[i,1] = 5 ∧ H[i,2] = 0 then P(Q(1,SQRT(H[i,3]),DI(H[i,3])) else
    P(Q(1,S(1,P(H[i,3],H[i,3])),DI(H[i,3]));

```

```

real procedure RIF(i); value i; integer i; RIF:=
if H[i,1] = -1 then VALUECN(i) else if i = x then x0 else if i = y then y0 else
if H[i,2] = 1 then (RIF(H[i,1]) + RIF(H[i,3])) else
if H[i,2] = 2 then (RIF(H[i,1]) * RIF(H[i,3])) else
if H[i,2] = 3 then (RIF(H[i,1]) / RIF(H[i,3])) else
if H[i,1] = 1 ^ H[i,2] = 0 then sin(RIF(H[i,3])) else
if H[i,1] = 2 ^ H[i,2] = 0 then cos(RIF(H[i,3])) else
if H[i,1] = 3 ^ H[i,2] = 0 then exp(RIF(H[i,3])) else
if H[i,1] = 4 ^ H[i,2] = 0 then ln(RIF(H[i,3])) else
if H[i,1] = 5 ^ H[i,2] = 0 then sqrt(RIF(H[i,3])) else arctan(RIF(H[i,3]));
procedure CZERO(x0,y0,r,s,e); real r,s,x0,y0; array e;
begin real a,b,c,d,e1,e2,e3,g,h,r0,r1,r2,s0,s1,s2,t,u;
  e1:=e[1]^2; e2:=e[2]^2; e3:=e[3]^2; a:=x0;
  g:=sqrt(a*a + y0^2)*.1 + .1; h:=0; c:=-.5; d:=0;
  x0:=a+g; r0:=r; s0:=s; x0:=a-g; r1:=r; s1:=s;
  r0:=r0-r1; s0:=s0-s1; x0:=a;
  r2:=r; s2:=s;
LL:  r1:=r1-r2; s1:=s1-s2;
  t:=r0*c - s0*d -r1; u:=r0*d +s0*c -s1;
  a:=(t-r1)*c-(u-s1)*d-r1; b:=(t-r1)*d +(u-s1)*c-s1;
  r0:=t*c-u*d; s0:=t*d+u*c;
  t:=-2.0*((1.0+c)*r2-d*s2); u:=-2.0*((1.0+c)*s2+d*r2);
  CSQRT(a*a-b*b+2.0*(t*r0-u*s0),2.0*(a*b+t*s0+u*r0),c,d);
  if a*c+b*d<0 then begin c:=-c; d:=-d end;
  a:=a+c; b:=b+d;
  c:=(a*t+b*u)/(a*a+b*b); d:=(a*u-b*t)/(a*a+b*b);
  a:=sqrt(c*c+d*d)/10;
  if a>1 then begin c:=c/a; d:=d/a end;
  a:=g*c-h*d; h:=g*d+h*c; g:=a; x0:=x0+g; y0:=y0+h;
  r0:=r1; s0:=s1; r1:=r2; s1:=s2; r2:=r; s2:=s;
  if g*g+h*h>(x0^2 + y0^2)*e1+e2 ^ r2*r2 + s2*s2>e3 then goto LL;
  X0:=x0; Y0:=y0
end  CZERO;
real procedure CSQRT(a,b,rp,ip); value a,b; real a,b,rp,ip;
begin rp:=sqrt((abs(a) +sqrt(a*a +b*b))/2.0);
  ip:=b/(2.0*rp);

```

```

    if a<0 then begin ip:=if b>0 then rp else -rp; rp:=abs(b) end;
    CSQRT:=rp
end CSQRT;
integer procedure f;
begin integer r;real r1,r2; switch F:=prod,som,quot,versch,sin,cos,exp,
    ln,sqrt,arctan,var,getal;r:=read;goto F[r];
prod:  f:=P(f,f); goto END;
som:   f:=S(f,f); goto END;
quot:  f:=Q(f,f); goto END;
versch: f:=D(f,f); goto END;
sin:   f:=SIN(f); goto END;
cos:   f:=COS(f); goto END;
exp:   f:=EXP(f); goto END;
ln:    f:=LN(f); goto END;
sqrt:  f:=SQRT(f); goto END;
arctan: f:=ARCTAN(f); goto END;
var:   f:=z; goto END;
getal: r1:=read;r2:=read;f:=NUMBER(STORECN(r1,r2));
END:   end;
BEGIN OF CALCN: k:=kc:=-1; NUMBER(STORECN(0,0)); NUMBER(STORECN(1,0));
    z:=STORE(-2,0,0); x:=STORE(-2,0,0); y:=STORE(-2,0,0);
    x0:=read; y0:=read; for i:=1,2,3 do e[i]:=read;
    RI(DI(f),R,I); CZERO(x0,y0,RIF(R),RIF(I),e); PUNLCR;
    FLOP(3,1,X0); PUTTEXT1(† , †); FLOP(3,1,Y0)
end
end

```

```

begin comment  Inverteren van reeks met complexe coeff. R1261 SJB 040166/2;
integer kmax,kcmax,m0; kmax:= XEEN(1023);
      kcmax:= XEEN(1023 × 1024): 1024;m0:=read;
begin integer k,kc,kf,z0,z,n,K,K1; real x0,y0; integer array c[2:m0],
d[1:m0], a[0:m0-2], H[0:kmax,1:3],h[0:m0-2], b[0:m0-2],
e[0:m0-2], g[0:m0-2]; array HC[0:kcmax,1:2],fac[1:m0];
integer procedure STORE(i,1,j); value i,j; integer i,1,j;
begin STORE:= k:= k + 1;
if k > kmax then begin PUTEXT1(k too large); stop end;
      H[k,1]:= i; H[k,2]:= 1; H[k,3]:= j
end;
integer procedure S(i,j); value i,j; integer i,j; S:=
if i = 0 then j else if j = 0 then i else STORE(i,1,j);
integer procedure D(i,j); integer i,j; D:= S(i,P(NUMBER(STORECN(-1,0)),j));
integer procedure P(i,j); value i,j; integer i,j; P:=
if i = 0 ∨ j = 0 then 0 else if i = 1 then j else
if j = 1 then i else STORE(i,2,j);
integer procedure Q(i,j); integer i,j; Q:= STORE(i,3,j);
integer procedure SIN(i); integer i; SIN:= STORE(1,0,i);
integer procedure COS(i); integer i; COS:= STORE(2,0,i);
integer procedure EXP(i); integer i; EXP:= STORE(3,0,i);
integer procedure LN(i); integer i; LN:= STORE(4,0,i);
integer procedure SQRT(i); integer i; SQRT:= STORE(5,0,i);
integer procedure ARCTAN(i); integer i; ARCTAN:= STORE(6,0,i);
integer procedure NUMBER(i); integer i; NUMBER:=STORE(-1,0,i);
integer procedure STORECN(a1,a2); real a1,a2;
begin STORECN:=kc:=kc+1;
if kc> kcmax then begin PUTEXT1(kc too large );stop end;
      HC[kc,1]:=a1; HC[kc,2]:=a2
end;
integer procedure pr(i,j); value i,j; integer i,j;
begin real b1,b2; b1:=HC[i,1]×HC[j,1] - HC[i,2]×HC[j,2];
      b2:=HC[i,1]×HC[j,2]+ HC[i,2]×HC[j,1];
      pr:=STORECN(b1,b2)
end;
integer procedure sc(i,j); value i,j; integer i,j;

```

```

begin real b1,b2; b1:=HC[i,1]+HC[j,1]; b2:=HC[i,2]+HC[j,2];
      so:= STORECN(b1,b2)
end;
integer procedure di(i,j); value i,j; integer i,j;
begin real b1,b2; b1:=HC[i,1]-HC[j,1]; b2:=HC[i,2]-HC[j,2];
      di:= STORECN(b1,b2)
end;
integer procedure qu(i,j); value i,j; integer i,j;
begin real b1,b2; b1:=(HC[i,1]×HC[j,1]-HC[i,2]×HC[j,2])/
      (HC[j,1]×HC[j,1]+HC[j,2]×HC[j,2]);
      b2:=(HC[i,2]×HC[j,1]-HC[i,1]×HC[j,2])/
      (HC[j,1]×HC[j,1]+HC[j,2]×HC[j,2]);
      qu:= STORECN(b1,b2)
end;
integer procedure ex(i); value i; integer i;
begin real b1,b2; b1:=exp(HC[i,1])×cos(HC[i,2]);
      b2:=exp(HC[i,1])×sin(HC[i,2]);
      ex:=STORECN(b1,b2)
end;
integer procedure lg(i); value i; integer i;
begin real b1,b2; b1:=ln(sqrt(HC[i,1]×HC[i,1]+HC[i,2]×HC[i,2]));
      b2:=arctan( HC[i,2]/HC[i,1] );
      lg:=STORECN(b1,b2)
end;
integer procedure sq(i); value i; integer i;
begin real b1,b2; b1:=sqrt(HC[i,1]×HC[i,1]+HC[i,2]×HC[i,2])×
      cos(arctan(HC[i,2]/HC[i,1])/2);
      b2:=sqrt(HC[i,1]×HC[i,1]+HC[i,2]×HC[i,2])×
      sin(arctan(HC[i,2]/HC[i,1])/2);
      sq:=STORECN(b1,b2)
end;
integer procedure si(i); value i; integer i;
begin real b1,b2; b1:=sin(HC[i,1])×(exp(HC[i,2])+exp(-HC[i,2]))/2 ;
      b2:=cos(HC[i,1])×(exp(-HC[i,2])-exp( HC[i,2]))/2;
      si:=STORECN(b1,b2)
end;

```

```

integer procedure co(i); value i; integer i;
begin real b1,b2; b1:=cos(HC[i,1])*(exp(HC[i,2])+exp(-HC[i,2]))/2 ;
      b2:=sin(HC[i,1])*(exp(-HC[i,2])-exp(HC[i,2]))/2 ;
      co:=STORECN(b1,b2)

end;

integer procedure ar(i); value i; integer i;
begin real b1,b2; b1:=(arctan(-HC[i,1]/(HC[i,2]-1))-arctan(-HC[i,1]/
      (HC[i,2]+1)))/2;
      b2:=-ln(sqrt(((1-HC[i,2])*(1-HC[i,2])+HC[i,1]*HC[i,1])/
      ((1+HC[i,2])*(1+HC[i,2])+HC[i,1]*HC[i,1])))/2 ;
      ar:=STORECN(b1,b2)

end;

integer procedure sum(i,g,b,ei); value g,b; integer i,g,b,ei;
begin real s; s:=0; for i:=g step 1 until b do s:=so(s,ei); sum:=s end;

integer procedure som(l,j); value l; integer l,j;
begin integer i; som:=
if l=1 then a[j] else sum(i,0,j,pr(a[i],som(l-1,j-i)))
end;

real procedure FAC(i); value i; integer i;
begin integer j; A: if i<kf then FAC:= fac[i] else
begin for j:=2 step 1 until m0 do fac[kf+j]:= fac[kf+j-1]*(kf+j);
kf:= kf+m0; goto A
end
end;

integer procedure DI(i); value i; integer i; DI:=
if H[i,1]=-1 then 0 else if i=z then 1 else
if H[i,2]=1 then S(DI(H[i,1]),DI(H[i,3])) else
if H[i,2]=2 then S(P(H[i,1],DI(H[i,3])),P(H[i,3],DI(H[i,1]))) else
if H[i,2]=3 then Q(D(P(H[i,3],DI(H[i,1])),P(H[i,1],DI(H[i,3])),
      P(H[i,3],H[i,3]))
      else
if H[i,1]=1 ^ H[i,2]=0 then P(COS(H[i,3]),DI(H[i,3])) else
if H[i,1]=2 ^ H[i,2]=0 then P(NUMBER(STORECN(-1,0)),
      P(SIN(H[i,3]),DI(H[i,3])))
      else
if H[i,1]=3 ^ H[i,2]=0 then P(EXP(H[i,3]),DI(H[i,3])) else

```

```

if H[i,1]≠4 ∧ H[i,2]≠0 then P(Q(1,H[i,3]),DI(H[i,3])) else
if H[i,1]≠5 ∧ H[i,2]≠0 then P(Q(1,SQRT(H[i,3])),DI(H[i,3])) else
P(Q(1,S(1,P(H[i,3],H[i,3])),DI(H[i,3]));
integer procedure VALUE(i); value i; integer i; VALUE:=
if H[i,1]≠-1 then H[i,3] else if i=z then z0 else
if H[i,2]≠1 then so(VALUE(H[i,1]),VALUE(H[i,3])) else
if H[i,2]≠2 then pr(VALUE(H[i,1]),VALUE(H[i,3])) else
if H[i,2]≠3 then qu(VALUE(H[i,1]),VALUE(H[i,3])) else
if H[i,1]≠1 ∧ H[i,2]≠0 then si(VALUE(H[i,3])) else
if H[i,1]≠2 ∧ H[i,2]≠0 then co(VALUE(H[i,3])) else
if H[i,1]≠3 ∧ H[i,2]≠0 then ex(VALUE(H[i,3])) else
if H[i,1]≠4 ∧ H[i,2]≠0 then lg(VALUE(H[i,3])) else
if H[i,1]≠5 ∧ H[i,2]≠0 then sq(VALUE(H[i,3])) else ar(VALUE(H[i,3]));
integer procedure f;
begin integer r;real r1,r2; switch F:=prod,som,quot,versch,sin,cos,exp,
ln,sqrt,arctan,var,getal;r:=read;goto F[r];
prod: f:=P(f,f); goto END;
som: f:=S(f,f); goto END;
quot: f:=Q(f,f); goto END;
versch: f:=D(f,f); goto END;
sin: f:=SIN(f); goto END;
cos: f:=COS(f); goto END;
exp: f:=EXP(f); goto END;
ln: f:=LN(f); goto END;
sqrt: f:=SQRT(f); goto END;
arctan: f:=ARCTAN(f); goto END;
var: f:=z; goto END;
getal: r1:=read;r2:=read;f:=NUMBER(STORECN(r1,r2));
END: end;
procedure SP;
begin integer i;
for n:=2 step 1 until m0 do c[n]:=qu(VALUE(d[n]),STORECN(FAC(n),0));
if HC[c[2],1]K10-10 ∧ HC[c[2],2]K10-10 then begin PUTTEXT1(⌘ apezadel ⌘); stop end;
b[0]:=VALUE(h[0]); for n:=1 step 1 until m0-2 do b[n]:=
qu(VALUE(h[n]),STORECN(FAC(n),0)); a[0]:=sq(qu(1,c[2])); e[0]:=b[0]; K:=kc;
for n:=1 step 1 until m0-2 do

```

```

begin a[n]:=qu(so(sum(i,3,n+2,pr(c[i],som(i,n-i+2))),
pr(c[2],sum(i,1,n-1,pr(a[i],a[n-i]))),pr(c[2],pr(STORECN(-2,0),a[0])));
HC[K+n,1]:=HC[a[n],1]; HC[K+n,2]:=HC[a[n],2]; kc:=K+n; a[n]:=K+n
end; K1:=K+m0-2;
for n:=1 step 1 until m0-2 do
begin e[n]:=sum(i,1,n,pr(b[i],som(i,n-i)));
HC[K1+n,1]:=HC[e[n],1]; HC[K1+n,2]:=HC[e[n],2]; kc:=K1+n; e[n]:=K1+n
end;
for n:=0 step 1 until m0-2 do
begin g[n]:=sum(i,0,n,pr(pr(STORECN(i+1,0),a[i]),e[n-i]));
PUNLCR; PUTEXT1(⌊ g[ ⌋); ABSFIXP(2,0,n); PUTEXT1(⌊ ⌋);
FLOP(3,1,HC[g[n],1]); PUTEXT1(⌊ , ⌋); FLOP(3,1,HC[g[n],2])
end
end;
BEGIN OF CALCN:k:=kc:=-1;kf:=0; fac[1]:=1;NUMBER(STORECN(0,0));
      NUMBER(STORECN(1,0)); z:=STORE(-2,0,0); x0:=read;
      y0:=read;z0:=STORECN(x0,y0);
      for n:=1 step 1 until m0 do
      begin if n=1 then d[1]:=DI(f) else d[n]:=DI(d[n-1]) end;
      for n:=0 step 1 until m0-2 do
      begin if n=0 then h[0]:=f else h[n]:=DI(h[n-1]) end; SP
end
end

```

References

1. H.A. Lauwerier Asymptotic Expansions
 Math. Centre tract 13
 Math. Centrum (1966)
2. R.P. v.d. Riet Algebraic operations in ALGOL 60
 The Cauchy-problem I
 Report TW 97, Math. Centrum (1965)
3. R.P. v.d. Riet An application of a method for algebraic manipulation
 in ALGOL 60
 Report TW 99, Math. Centrum (1966)